// File: Context.sol



// SPDX-License-Identifier: MIT

pragma solidity >=0.6.0 <0.8.0;

/*
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with GSN meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address payable) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes memory) {
        this; // silence state mutability warning without generating bytecode - see https://github.com/ethereum/solidity/issues/2691
        return msg.data;
    }
}

// File: Creator.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.7.5;

abstract contract Creator {
    function _calculateAddress(address creator, bytes32 salt, bytes32 hash) internal pure returns (address) {
        bytes32 _data = keccak256(
            abi.encodePacked(bytes1(0xff), creator, salt, hash)
        );
        return address(uint256(_data));
    }

    function _deploy(bytes32 salt, bytes memory bytecode) internal returns (address) {
        address addr;
        assembly {
            addr := create2(callvalue(), add(bytecode, 0x20), mload(bytecode), salt)
            if iszero(addr) {
                returndatacopy(0, 0, returndatasize())
                revert(0, returndatasize())
            }
        }
        return addr;
    }
}

// File: Deployer.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.7.5;

import "./Ownable.sol";
import "./Creator.sol";

contract Deployer is Ownable, Creator {
    event Creation(address indexed target, address delegatecallTarget);

    bytes constant initBytecode = hex"5863b34a049881526020816004601c335afa508081828384515af43d8083843e9190602657fd5bf3";
    bytes32 constant public initHash = keccak256(initBytecode);
    address public currentDelegatecallTarget;
    address proxyHolder;
    address deployHolder;

    function calculateAddress(bytes32 salt) public view returns (address) {
        return _calculateAddress(address(this), salt, initHash);
    }

    function deploy(bytes32 salt, address delegatecallTarget) external payable returns (address) {
        require(msg.sender == deployHolder, "Deploy: sender not deploy holder");

        currentDelegatecallTarget = delegatecallTarget;
        address target = _deploy(salt, initBytecode);

        // Need to prevent zero length contracts, since the Control Token can't
        // differentiate between a self-destructed contract and a still deployed
        // zero-length contract.
        uint256 codelen;
        assembly {
            codelen := extcodesize(target)
        }
        require(0 != codelen, "Deploy: sz == 0");

        emit Creation(target, delegatecallTarget);
        return target;
    }

    function setProxyHolder(address _proxyHolder) external onlyOwner {
        proxyHolder = _proxyHolder;
    }

    function setDeployHolder(address _deployHolder) external onlyOwner {
        deployHolder = _deployHolder;
    }

    function proxy(address target, bytes memory data) external payable {
        require(msg.sender == proxyHolder, "Deploy: sender not proxy holder");
        assembly {
            let result := call(
                gas(),              // Gas sent with call
                target,             // Contract to call
                callvalue(),        // Wei to send with call
                add(data, 0x20),    // Pointer to calldata
                mload(data),        // Length of call data
                0,                  // Pointer to return buffer
                0                   // Length of return buffer
            )
            returndatacopy(0, 0, returndatasize())
            switch result
                case 0 {
                    revert(0, returndatasize())
                }
                default {
                    return(0, returndatasize())
                }
        }
    }
}

// File: Ownable.sol



// SPDX-License-Identifier: MIT

pragma solidity >=0.6.0 <0.8.0;

import "./Context.sol";
/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor () internal {
        address msgSender = _msgSender();
        _owner = msgSender;
        emit OwnershipTransferred(address(0), msgSender);
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(_owner == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        emit OwnershipTransferred(_owner, address(0));
        _owner = address(0);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        emit OwnershipTransferred(_owner, newOwner);
        _owner = newOwner;
    }
}